Beispiel zur linearen Rueckfuehrregelung
(zu Abschnitt 3.2.2, nach Siegert)

1. Das Modell - die ODE

Der Einfachheit halber bewegen wir uns im Eindimensionalen, also n=1. Dafuer waehlen wir aber ein System, das mittels einer ODE 2. Ordnung (also mit zweiter Zeitableitung) beschrieben wird.
x(t): Zustandsgroesse des Systems
u(t): Stellgroesse des Reglers
k: Rueckfuehrkonstante (beschreibt linearen Einfluss der Stellgroesse auf die Zustandsgroesse und somit auf den Prozess)
x_soll: Soll-Wert fuer x(t)
y(t): Abweichung des Ist-Zustands vom Soll-Zustand, also y(t)=x(t)-x_soll

```
> restart;
> with(LinearAlgebra):
> with(plots):
> with(DEtools):
> readlib(unassign);
```

proc( ) **description** "remove an assignment from an assigned expression"; ... **end proc** **(1.1)**

```
> ODE_1:=diff(diff(x(t),t),t)+a*diff(x(t),t)+b*x(t)+c+k*u(t)=0;
```

$$ODE\_1 := \frac{d^2}{dt^2}x(t) + a\left(\frac{d}{dt}x(t)\right) + b\,x(t) + c + k\,u(t) = 0 \quad \textbf{(1.2)}$$

Hierbei handelt es sich um eine lineare ODE mit konstanten Koeffizienten a, b und c (also etwas bedeutend Einfacheres als die in Kapitel 3.1 betrachteten ODE mit allgemeinerer rechter Seite!).

2. Loesung ohne Regelung

Zunaechst versuchen wir unser Glueck ohne Regelung und loesen die Differentialgleichung direkt. Unser Ziel ist dabei (zumindest asymptotisch) ein stabiler (stationaerer) Systemzustand x(t), also ohne aenderungen von x(t) in der Zeit.

```
> k:=0;
```

$$k := 0 \quad \textbf{(2.1)}$$

```
> ODE_1;
```

$$\frac{d^2}{dt^2}x(t) + a\left(\frac{d}{dt}x(t)\right) + b\,x(t) + c = 0 \quad \textbf{(2.2)}$$

```
> SOL_1:=dsolve(ODE_1,x(t));
```

$$SOL\_1 := x(t) = e^{\left(-\frac{1}{2}a + \frac{1}{2}\sqrt{a^2-4b}\right)t}\,\_C2 + e^{\left(-\frac{1}{2}a - \frac{1}{2}\sqrt{a^2-4b}\right)t}\,\_C1 - \frac{c}{b} \quad \textbf{(2.3)}$$

Interpretation:
Die Loesung ist nicht eindeutig bestimmt, sondern ueber die beiden freien Parameter _C1 und _C2 parametrisiert. Fixiert werden koennen diese durch Wahl von Anfangswerten (zwei, da es sich um eine ODE 2. Ordnung handelt, also z.B. die Werte von x und seiner Ableitung an der Stelle t=0). Stationaere Grenzloesung ist uebrigens x(t)=-c/b; wenn x(t) also in einen stabilen

Systemzustand einschwingt, so muss es dieser sein.

Im Falle von negativem a droht Divergenz: Mindestens einer der Exponentialterme hat einen positiven Realteil im Exponenten, der i.A. (ausser bei Vorfaktor 0) zu exponentiellem Wachstum fuehrt.

Im Falle eines negativen Werts unter der Wurzel kommt es zu einem Imaginaerteil im Exponenten, also zu ueberlagerten Oszillationen.

Die Konvergenz in einen stabilen (stationaeren) Zustand ist also bei beliebiger Lage der Modellparameter a,b und c sowie der (aus den Anfangswerten folgenden) freien Parameter _C1 und _C2 keineswegs gesichert! Daraus resultiert der Wunsch bzw. die Notwendigkeit nach Regelung.

3. Zuschalten der Regelung

Jetzt schalten wir die Stellgroesse u(t) des Reglers zu und spezifizieren sie genauer (allgemeiner PID-Regler):

```
> unassign('k');
> u(t):=K_P*(x(t)-x_soll)+K_D*diff(x(t),t)+K_I*int((x(z)-
  x_soll),z=0..t);
```

$$u(t) := K\_P\,(x(t) - x\_soll) + K\_D\left(\frac{d}{dt}\,x(t)\right) + K\_I\left(\int_0^t (x(z) - x\_soll)\,dz\right) \qquad (3.1)$$

Ausserdem gehen wir von der Zustandsgroesse x(t) ueber zur Soll-Ist-Abweichung y(t).

```
> x(t):=y(t)+x_soll;
```

$$x(t) := y(t) + x\_soll \qquad (3.2)$$

```
> simplify(ODE_1);
```

$$\frac{d^2}{dt^2}\,y(t) + a\left(\frac{d}{dt}\,y(t)\right) + b\,y(t) + b\,x\_soll + c + k\,K\_P\,x(t) - k\,K\_P\,x\_soll + k\,K\_D\left(\frac{d}{dt}\,x(t)\right) + k\,K\_I\left(\int_0^t (x(z) - x\_soll)\,dz\right) = 0 \qquad (3.3)$$

```
> abk:=A=a+k*K_D, B=b+k*K_P, C=c+b*x_soll, E=k*K_I;
```

$$abk := A = a + k\,K\_D,\ B = b + k\,K\_P,\ C = c + b\,x\_soll,\ E = k\,K\_I \qquad (3.4)$$

Mit diesen Abkuerzungen laesst sich obige ODE schreiben als

```
> ODE_2:=diff(diff(y(t),t),t)+A*diff(y(t),t)+B*y(t)+C+E*int(y
  (z),z=0..t)=0;
```

$$ODE\_2 := \frac{d^2}{dt^2}\,y(t) + A\left(\frac{d}{dt}\,y(t)\right) + B\,y(t) + C + E\left(\int_0^t y(z)\,dz\right) = 0 \qquad (3.5)$$

Man beachte:
* Die Parameter A und B vor der ersten Ableitung von y(t) bzw. vor y(t) sind durch die Regelparameter K_D bzw. K_P beeinfllussbar.

* Der konstante Parameter C ist allein durch das System vorgegeben.
* Der Parameter E vor dem Integral ueber y(t) ist nur im Fall eines Regler-Integralteils ungleich Null.

4. Loesung ohne Integral-Komponente (K_I=0)

Zunaechst lassen wir die Integral-Komponente weg, waehlen also K_I=E=0.

```
> E:=0;
```

$$E := 0 \tag{4.1}$$

```
> ODE_2;
```

$$\frac{d^2}{dt^2} y(t) + A \left(\frac{d}{dt} y(t) \right) + B\, y(t) + C = 0 \tag{4.2}$$

```
> SOL_2:=dsolve(ODE_2,y(t));
```

$$SOL\_2 := y(t) = e^{\left(-\frac{1}{2} A + \frac{1}{2}\sqrt{A^2 - 4B}\right) t}\, \_C2 + e^{\left(-\frac{1}{2} A - \frac{1}{2}\sqrt{A^2 - 4B}\right) t}\, \_C1 - \frac{C}{B} \tag{4.3}$$

Interpretation:
Die Loesung ist wieder nicht eindeutig gegeben, sondern ueber die freien Parameter _C1 und _C2 parametrisiert. Zur Fixierung dienen wie zuvor die Anfangswerte.

Die ODE sieht genau gleich aus wie zuvor ohne Regelung, nur dass die Koeffizienten jetzt mit Grossbuchstaben bezeichnet sind. Und genau da liegt der feine Unterschied.
Im Gegensatz zu vorhin, als die Parameter a, b und c reine Modellparameter waren, hat man ueber die Regelung jetzt Einfluss auf die Parameter A und B. Eine gute Regelung wird die (vorher unvermeidlichen) Problemfaelle, in denen keine stationaere Loesung erreicht wird, zu umgehen suchen.

Im Idealfall rascher Konvergenz gegen eine stationaere Loesung ohne Oszillationen muss gelten:
1. A positiv und moeglichst gross (negativ exponentielles Abklingen)
2. Term unter der Wurzel nicht-negativ (kein Imaginaerteil der Wurzel, keine Schwingungen), moeglichst nahe Null (gut fuer's rasche Abklingen)

Folgerung: Nur mit K_P (reiner P-Regler) kann A nicht eingestellt werden - ein P-Regler reicht im Allgemeinen in dieser Situation also nicht aus!

Beispiele:

```
> _C1:=1; _C2:=1;
> C:=1; B:=2;
```

$$\_C1 := 1$$

$$\_C2 := 1$$

$$C := 1$$

$$B := 2 \tag{4.4}$$

```
> A:=0;  plot(op(2,SOL_2),t=0..10);
```

$A := 0$



_C1 und C_2 seien durch die Anfangsbedingungen so gegeben. B und C moegen aus den Modellgroessen a,b,c und x_soll sowie aus der Wahl der Regelparameter K_D und K_P resultieren. Ferner sei alles so gewaehlt, dass A=0 gilt. Es ergibt sich:
Realteil im Exponent Null wegen A=0, somit weder Abklingen noch Aufschaukeln, aber Oszillation - ein solches A muss durch einen geeigneten Regelparameter K_D vermieden werden!

```
> A:=-5;  plot(op(2,SOL_2),t=0..10);
```

$A := -5$



Jetzt Imaginaerteil im Exponent Null (Term unter Wurzel positiv), positiver Exponent, somit exponentielles Wachstum. Auch ein solches A fuehrt nicht in einen stabilen Systemzustand und ist durch Regelung zu vermeiden.

```
> A:=5;  plot(op(2,SOL_2),t=0..10);
```

$$A := 5$$



Hier gibt es zwar eine stationaere Loesung y=-C/B=-0.5 (insofern ist unsere Regelung zumindest ein Teilerfolg), diese ist aber nicht der Sollwert Null (was ja auch nicht sein kann, wenn C nicht zufaellig 0 ist)! Wir wollen aber die stationaere Loesung y=0; schliesslich gibt y(t) a die Abweichung der Zustandsgroesse x(t) von ihrem Sollwert x_soll an.
Man scheint also eine Integral-Komponente zu benoetigen!

5. Loesung mit Integral-Komponente (K_I ungleich Null)

Also schalten wir nun den kompletten PID-Regler zu.

```
> unassign('A','B','C','E');
> ODE_2;
```

$$\frac{d^2}{dt^2}\,y(t) + A\left(\frac{d}{dt}\,y(t)\right) + B\,y(t) + C + E\left(\int_0^t y(z)\,dz\right) = 0 \qquad (5.1)$$

Diese ODE 2. Ordnung ist gar keine reinrassige ODE mehr - sie hat zusaetzlich zu den Ableitungen auch einen *Integralterm.* Dennoch kann diese Gleichung ebenfalls mit einem

Exponentialansatz y(t)=f(C)*exp(-lambda*t) geloest werden (Laplace-Transformation).
Wir lassen wieder Maple arbeiten:

```
> SOL_3:=dsolve(ODE_2,y(t),method=laplace);
```

$$SOL\_3 := y(t) = \sum_{\_\alpha = RootOf(\_Z^3 + A\,\_Z^2 + B\,\_Z + E)} \frac{e^{\_\alpha\, t}\,(-C + \_\alpha\,(D(y)(0) + (\_\alpha + A)\,y(0)))}{3\,\_\alpha^2 + 2\,\_\alpha\, A + B} \qquad (5.2)$$

Ein ziemliches Monster, bei genauerem Hinsehen aber gar nicht so schlimm: eine Summe ueber verschiedene Exponentialfunktionen, wobei der Summationsparameter _alpha im Exponenten ueber alle Wurzeln des angegebenen Polynoms dritten Grades laeuft. Wie wir schon gesehen haben, muessen die Parameter A, B und E so gewaehlt werden, dass alle dieser Wurzeln negativen Realteil (im Idealfall auch verschwindenden Imaginaerteil) haben, wenn y(t) gegen Null und somit x(t) gegen x_soll konvergieren soll.

```
> EQU_1:=z^3+A*z^2+B*z+E=0;
```

$$EQU\_1 := z^3 + A\,z^2 + B\,z + E = 0 \qquad (5.3)$$

```
> solve(EQU_1,z);
```

$$\frac{1}{6}\left(36\,B\,A - 108\,E - 8\,A^3 + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A^3}\right)^{1/3}$$

$$- \frac{6\left(\frac{1}{3}\,B - \frac{1}{9}\,A^2\right)}{\left(36\,B\,A - 108\,E - 8\,A^3 + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A^3}\right)^{1/3}}$$

$$- \frac{1}{3}\,A,\; -\frac{1}{12}\left(36\,B\,A - 108\,E - 8\,A^3\right.$$

$$\left. + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A^3}\right)^{1/3}$$

$$+ \frac{3\left(\frac{1}{3}\,B - \frac{1}{9}\,A^2\right)}{\left(36\,B\,A - 108\,E - 8\,A^3 + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A^3}\right)^{1/3}}$$

$$- \frac{1}{3}\,A + \frac{1}{2}\,\mathrm{I}\sqrt{3}\left(\frac{1}{6}\left(36\,B\,A - 108\,E - 8\,A^3\right.\right.$$

$$\left. + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A^3}\right)^{1/3}$$

$$+ \left(6\left(\frac{1}{3}\,B - \frac{1}{9}\,A^2\right)\right) \Big/ \left(36\,B\,A - 108\,E - 8\,A^3 + 12\sqrt{12\,B^3 - 3\,B^2 A^2 - 54\,B\,A\,E + 81\,E^2 + 12\,E\,A\right.$$

$$\left.{}^{/3}\right),\; -\frac{1}{12}\left(36\,B\,A - 108\,E - 8\,A^3\right.$$

$$+ 12 \sqrt{12 B^3 - 3 B^2 A^2 - 54 B A E + 81 E^2 + 12 E A^3}\Big)^{1/3}$$

$$+ \frac{3 \left(\frac{1}{3} B - \frac{1}{9} A^2\right)}{\left(36 B A - 108 E - 8 A^3 + 12 \sqrt{12 B^3 - 3 B^2 A^2 - 54 B A E + 81 E^2 + 12 E A^3}\right)^{1/3}}$$

$$- \frac{1}{3} A - \frac{1}{2} \mathrm{I} \sqrt{3} \Bigg(\frac{1}{6} \Big(36 B A - 108 E - 8 A^3$$

$$+ 12 \sqrt{12 B^3 - 3 B^2 A^2 - 54 B A E + 81 E^2 + 12 E A^3}\Big)^{1/3}$$

$$+ \left(6 \left(\frac{1}{3} B - \frac{1}{9} A^2\right)\right) \Big/ \Big(36 B A - 108 E - 8 A^3 + 12 \sqrt{12 B^3 - 3 B^2 A^2 - 54 B A E + 81 E^2 + 12 E A}$$

$$/3 \Bigg)$$

Man analysiere diese Formeln und erinnere sich dabei an die Analysis: eine dieser drei Wurzeln ist reell, und die beiden anderen sind aufgrund der reellen Koeffizienten des Polynoms konjugiert komplex. Man kann die Nullstellen also schreiben z1=x, z2=u+iv, z3=u-iv mit den drei Parametern x, u und v. Somit muss man diese geeignet waehlen, daraus A, B und E bestimmen, und daraus wiederum die Werte fuer die Regelparameter K_P, K_D und K_I ermitteln.

Ein Beispiel:

Wir geben die Startwerte vor und den (fuer die Wurzeln) irrelevanten Parameter C (der ergibt sich ja auch zwingend aus den Modellparametern b,c und x_soll). Anschliessend geben wir A, B und E vor - in der Hoffnung, dass mit den Wurzeln nichts anbrennt:

```
> y(0):=0; D(y)(0):=0; C:=1;
```

$$y(0) := 0$$

$$D(y)(0) := 0$$

$$C := 1 \qquad \textbf{(5.5)}$$

```
> SOL_3;
```

$$y(t) = \sum_{\_\alpha = RootOf(\_Z^3 + A\_Z^2 + B\_Z + E)} \left(- \frac{e^{\_\alpha t}}{3\_\alpha^2 + 2\_\alpha A + B} \right) \qquad \textbf{(5.6)}$$

```
> A:=2; B:=3; E:=2;
```

$$A := 2$$

$$B := 3$$

$$E := 2 \qquad \textbf{(5.7)}$$

```
> SOL_3;
```

$$y(t) = \sum_{\_\alpha = RootOf(\_Z^3 + 2\_Z^2 + 3\_Z + 2)} \left(- \frac{e^{\_\alpha t}}{3\_\alpha^2 + 4\_\alpha + 3} \right) \quad (5.8)$$

```
> evalf(solve(EQU_1,z));
```

$$-1., -0.5000000000 + 1.322875656\,I, -0.5000000000 - 1.322875656\,I \quad (5.9)$$

Glueck gehabt - alle Wurzeln haben negativen Realteil!

```
> evalf(SOL_3);
```

$$y(t) = -0.5000000000\,e^{-1.t} + (0.2500000000 - 0.09449111825\,I)\,e^{(-0.5000000000 - 1.322875656\,I)\,t} + (0.2500000000 + 0.09449111825\,I)\,e^{(-0.5000000000 + 1.322875656\,I)\,t} \quad (5.10)$$

```
> plot(op(2,SOL_3),t=0..10);
```



Man sieht: Der Parametersatz fuehrt zu einer stationaeren Loesung. Die Oszillationen (es gab komplexe Nullstellen) trueben das Bild kaum. Schliesslich berechnen wir unsere Regelparameter K_P aus B=b+k*K_P, K_D aus A=a+k*K_D und K_I aus E=k*K_I und haben somit unsere zum Ziel fuehrende Regelung im Griff.

Das Beispiel zeigt, dass die Regelungstechnik beherrschbar ist; es zeigt aber auch, dass bereits bei relativ einfachen Problemen wie den hier diskutierten ein erheblicher Modellaufwand zu treiben ist. Die Frage nach einfacher zugaenglichen Modellen und einer somit intuitiveren Simulation bzw. Regelung stellt sich deshalb hier besonders stark.